

Tema 6

Funcions reals de variable real.

Representació gràfica. Límits i continuïtat

Objectius

1. Definir funcions amb wxMaxima. Càlculs amb funcions.
2. Realitzar operacions amb funcions.
3. Representar funcions gràficament.
4. Calcular límits de funcions.
5. Estudiar la continuïtat de funcions.
6. Aplicació pràctica de les propietats de les funcions contínues.

Continguts

- 06-1. Definició de funcions. Exemples. Operacions amb funcions.
- 06-2. Representació gràfica de funcions. Representació gràfica de corbes planes.
- 06-3. Límits de funcions. Càlcul i propietats.
- 06-4. Funcions contínues. Estudi de la continuïtat.
- 06-5. Propietats de les funcions contínues. Aplicacions.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- AP86 APOSTOL, T.M. (1986)
Análisis Matemático

- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable
- GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones
- GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.
- SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

06-1.- Definició de funcions. Exemples. Operacions amb funcions

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_06-1.wxm**.

Recordem que s'anomena funció real de variable real a una terna (A, B, f) en la qual A, B són subconjunts (no buits) de nombres reals i f és una aplicació de A en B . El conjunt A s'anomena domini o camp d'existència de la funció i el conjunt B s'anomena recorregut. Habitualment s'escriu $f: A \subset \mathbb{R} \rightarrow B$ per designar una funció f de A en B . Quan en endavant diguem funció, ens referirem a una funció real de variable real.

Una funció s'acostuma a definir mitjançant una expressió analítica que indica com es transforma l'element (antiimatge) del conjunt A per obtenir l'element (imatge) del conjunt B que li correspon. Per exemple: $f(x) = 1 + 3x - 2x^2$, $x \geq 0$ seria la forma de donar una funció definida mitjançant l'expressió analítica indicada (polinomi) i de camp d'existència $A = [0, +\infty[$.

Per definir una funció amb wxMaxima, assignar un símbol a la funció, a continuació un parèntesi amb el símbol designat a la variable, a continuació els símbols $:=$ i finalment l'expressió analítica de la funció. Per exemple:

```
(%i1) f(x):=1+3*x-2*x^2;  
(%o1) f(x):= 1 + 3 x + (-2) x^2
```

Això indica que aquesta funció es representa amb el símbol f i que la seva variable (és a dir, un punt qualsevol del seu camp d'existència) es representa per x .

Una manera alternativa de definir una funció (que més endavant veurem que és molt útil) és amb la instrucció “define”:

```
(%i2) define(f(x), 1+3*x-2*x^2 );  
(%o2) f(x):= -2 x^2 + 3 x + 1
```

Altres exemples de funcions:

```
(%i3) f1(x):=2*cos(3*x)+4*sin(2*x);  
      f2(t):=1/(1+t^2);  
      f3(u):=log(u+1);  
(%o3) f1(x):= 2 cos(3 x) + 4 sin(2 x)  
(%o4) f2(t):=  $\frac{1}{1+t^2}$   
(%o5) f3(u):= log(u + 1)
```

Es pot obtenir la imatge d'un punt del domini amb una sintaxi prou senzilla; per exemple:

```
(%i6) f(0);          f(2.34);
(%o6) 1
      (%o7) -2.931199999999999
(%i8) f1(%pi);      f1(2.3456);      f2(3.457);
(%o8) -2
      (%o9) -2.540661729040671
      (%o10) 0.077215015015618
(%i11) f3(%e);      f3(2);          f3(2), numer;
(%o11) log(%e + 1)
      (%o12) log(3)
      (%o13) 1.09861228866811
```

S'observa que en alguns casos sembla que la imatge “es deixi indicada” sense calcular el valor numèric; el que passa és que no es dona aquest valor expressament a menys que s'indiqui que es vol la representació decimal de la imatge o es doni l'antiimatge en representació de coma flotant. També es pot obtenir el valor d'una funció en un punt assignant un valor a la variable de la funció (que més endavant caldrà eliminar si es vol seguir treballant amb la funció). Per exemple:

```
(%i14) x:5.23;      t:4.321;      u:6.543;
      f(x);          f2(t);          f3(u);
(%o14) 5.23
      (%o15) 4.321
      (%o16) 6.543
      (%o17) -38.015800000000001
      (%o18) 0.050836150460975
      (%o19) 2.020619980871494
```

Cal recordar la notació de les funcions elementals predefinides a wxMaxima (vegeu la Taula 02-1 del Tema 2).

Operacions algebraiques amb funcions. Per efectuar operacions algebraiques (suma, diferència, producte per un escalar, producte i quocient), la sintaxi és prou senzilla. Per exemple:

- Suma i diferència:

```
(%i1) f1(x):=2*cos(3*x)+4*sin(2*x)$      f2(x):=1/(1+x^2)$      f3(x):=log(x+1)$
```

```
(%i4) f10(x):=f1(x)+f3(x);          'f10(x)=f10(x);
(%o4) f10(x):= f1(x)+ f3(x)
(%o5) f10(x)=log(x+1)+2 cos(3 x)+4 sin(2 x)
(%i6) f11(x):=f1(x)-f2(x);          'f11(x)=f11(x);
(%o6) f11(x):= f1(x)- f2(x)
(%o7) f11(x)=2 cos(3 x)+4 sin(2 x)- $\frac{1}{x^2+1}$ 
```

- Producte per un escalar i combinació lineal:

```
(%i8) f12(x):=3.456*f1(x);          'f12(x)=f12(x);
      f13(x):=1.234*f1(x)-3.456*f3(x); 'f13(x)=f13(x);
(%o8) f12(x):= 3.456 f1(x)
(%o9) f12(x)=3.456 (2 cos(3 x)+4 sin(2 x))
(%o10) f13(x):= 1.234 f1(x)-3.456 f3(x)
(%o11) f13(x)=1.234 (2 cos(3 x)+4 sin(2 x))-3.456 log(x+1)
```

- Producte i quocient:

```
(%i12) f14(x):=f1(x)*f2(x);          'f14(x)=f14(x);
(%o12) f14(x):= f1(x) f2(x)
(%o13) f14(x)= $\frac{2 \cos(3 x)+4 \sin(2 x)}{x^2+1}$ 
(%i14) f15(x):=f1(x)/f2(x);          'f15(x)=f15(x);
(%o14) f15(x):= $\frac{f1(x)}{f2(x)}$ 
(%o15) f15(x)=(x2+1)(2 cos(3 x)+4 sin(2 x))
```

Composició de funcions. La sintaxi per a la composició de funcions en wxMaxima, que es pot veure a continuació, no fa servir el signe habitual de composició, sinó que s'escriu la imatge com “funció de la funció”. Observeu que es pot fer la composició de més de dues funcions:

```
(%i1) f1(x):=2*x+5$      f2(x):=sqrt(3+x^2)$      f3(x):=sin(x)$
(%i4) F1(x):=f1(f2(x));  'F1(x)=F1(x);
(%o4) F1(x):= f1( f2(x))
(%o5) F1(x)=2  $\sqrt{x^2+3}$  +5
```

```
(%i6) F2(x):=f2(f1(x));          'F2(x)=F2(x);
(%o6) F2(x):=f2(f1(x))
(%o7) F2(x)=sqrt((2 x + 5)^2 + 3)
(%i8) F3(x):=f1(f2(f3(x)));      'F3(x)=F3(x);
(%o8) F3(x):=f1(f2(f3(x)))
(%o9) F3(x)=2*sqrt(sin(x)^2 + 3) + 5
```

L'exemple serveix addicionalment per recordar que la composició de funcions no compleix la propietat commutativa.

Funcions definides a trossos o intervals. Es poden considerar i treballar amb funcions definides a trossos o intervals. La metodologia consisteix en designar amb una referència cada una de les diferents expressions i després organitzar les alternatives en una instrucció. Així, per exemple, la funció definida per:

$$F_1(x) = \begin{cases} 2x + 1, & \text{si } x < 0 \\ \sqrt{1 + x^2}, & \text{si } x \geq 0 \end{cases}$$

es defineix en wxMaxima de la manera següent:

```
(%i1) f1(x):=2*x+1;          f2(x):=sqrt(1+x^2);
      F1(x):= if x<0 then f1(x) else f2(x);
(%o1) f1(x):= 2 x + 1
(%o2) f2(x):=sqrt(1 + x^2)
(%o3) F1(x):=if x<0 then f1(x) else f2(x)
```

Si la funció està definida en tres intervals la sintaxi és igualment senzilla; per exemple, la funció:

$$G_1(x) = \begin{cases} e^{x/2}, & x \leq 0 \\ 1 + x \log(x/2), & 0 < x < 2 \\ e^{2-x}, & x \geq 2 \end{cases}$$

es defineix en wxMaxima de la manera següent:

```
(%i4) g1(x):=exp(x/2)$  g2(x):=1+x*log(x/2)$  g3(x):=exp(2-x)$
(%i7) G1(x):= if x<=0 then g1(x) else (if x<=2 then g2(x) else g3(x));
(%o7) G1(x):=if x<=0 then g1(x) else if x<=2 then g2(x) else g3(x)
```

06-2.- Representació gràfica de funcions. Representació gràfica de corbes planes.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_06-2.wxm**.

La representació gràfica de funcions es duu a terme amb l'opció “Gráficos” del menú principal. Abans de fer la representació gràfica cal definir la funció i després anar a “Plot” i decidir per una de les opcions que apareixen: Gráficos2D, Gráficos3D, Formato de gráficos. En el cas de funcions reals de variable real cal anar a Gráficos2D; el Formato de gráficos es recomana deixar-lo amb les opcions per defecte. El quadre de diàleg de Gráficos2D és el representat a la Figura 06-1.



Figura 06-1. Quadre de diàleg per a gràfics en 2D.

Comentem les indicacions i consells per complimentar el quadre de diàleg:

- A l'opció “Expresión(es)” cal escriure el símbol assignat a la funció o funcions que es volen representar, ja que es poden incloure diverses funcions en un mateix gràfic.
- A l'opció “Variable” cal especificar el rang de la variable; es poden fer proves per obtenir la zona de gràfic més adequada per a la representació que es vol obtenir. No cal, si no es vol, especificar el rang de les ordenades.
- A l'opció “Graduaciones” és per definir l'escala dels eixos.
- A l'opció “Formato” es pot optar per “predeterminado” o “en línea”. La primera, que es pot executar amb la instrucció “plot(...)”, mostra un gràfic en format anomenat “gnuplot” en el mateix document de wxMaxima, que apareix quan s'executa la instrucció i desapareix en executar la instrucció següent; en aquest gràfic, anant a la part superior i fent servir el botó dret del ratolí, s'accedeix a un menú de modificació de les opcions del gràfic que permet personalitzar-lo. En la segona opció, més recomanable si es vol mantenir la gràfica visible, s'inclou el gràfic enganxat a la sortida i amb el botó dret del ratolí es pot copiar i enganxar després en un altre document. L'opció de la gràfica “en línea” es pot executar amb la instrucció “wxplot(...)”. Hi ha una opció addicional que és “openmath” que permet editar i manipular el gràfic, a més de guardar-lo en un fitxer.

- A “Opciones” hi ha les opcions de gràfic: setzeroaxis, set size ratio 1, set grid. Aquestes opcions afecten el dibuix dels eixos de coordenades, la mida relativa de les escales en els eixos i al dibuix d’una malla.
- L’opció “Gráfico al archivo” permet exportar el gràfic a un fitxer que en el botó que hi ha més a la dreta permet especificar el nom del fitxer i la carpeta de destí. El format dels gràfics és postscript encapsulat, eps. Cal tenir doncs un programa que permeti obrir aquest tipus de fitxer i, si s’escau, modificar després el format gràfic.


Vegem amb un exemple com afecten les opcions al resultat que dona el programa. S’observarà que en tots els casos es fa servir la instrucció “wxplot(...)” que permet tenir la gràfica “enganxada” al mateix fitxer de wxMaxima. L’aspecte de la gràfica es pot adaptar d’alguna manera a les preferències o gustos de l’usuari i a les necessitats objectives de la representació gràfica.

```
(%i1) f(x):=x^3-2*x^2+4*x+4;    g(x):=1+x^2;
```


```
(%o1) f(x):=x^3-2*x^2+4*x+4
```

```
(%o2) g(x):=1+x^2
```

```
(%i3) wxplot2d([f(x)], [x,-4,4])$
```




```
(%i4) wxplot2d([g(x)], [x,-4,4])$
```




(%i5) wxplot2d([g(x)], [x,-4,4],
[gnuplot_preamble, "set size ratio 1"])\$

(%t5)




(%i6) wxplot2d([g(x)], [x,-4,4],
[gnuplot_preamble, "set size ratio 1; set grid"])\$

(%t6)



(%i7) wxplot2d([g(x)], [x,-4,4],
[gnuplot_preamble, "set size ratio 1; set grid"] , [style, [lines, 2]])\$


(%t7)



```
(%i8) wxplot2d([g(x)], [x,-4,4], [y, 0, 8],
               [gnuplot_preamble, "set size ratio 1; set grid; "], [style, [lines, 2]])$
```

plot2d: some values were clipped.


(%t8)



Es poden incloure en un mateix gràfic la representació de dues funcions, com es pot veure a continuació. Per fer-ho, cal simplement escriure a “Expresión(es)” els símbols de les dues funcions, separades per una coma.

```
(%i9) wxplot2d([f(x),g(x)], [x,-4,4],
               [gnuplot_preamble, "set size ratio 1; set grid; "], [legend, "f(x)", "g(x)"], [style, [lines, 2]])$
```

(%t9)



Les funcions definides a intervals també es poden representar gràficament. Per exemple, si es vol obtenir la representació gràfica de la funció definida per:

$$F_1(x) = \begin{cases} 2x+1, & \text{si } x < 0 \\ \sqrt{1+x^2}, & \text{si } x \geq 0 \end{cases}$$


Primer cal definir les expressions de la funció en cada un dels intervals, a continuació establir cada una de les expressions associada a l'interval corresponent i finalment executar la instrucció de representació gràfica tal com s'ha indicat. Així, en aquest cas, seria:

```
(%i1) f1(x):=2*x+1$
      f2(x):=sqrt(1+x^2)$
      F1(x):= if x<0 then f1(x) else f2(x);
(%o3) F1(x):=if x<0 then f1(x) else f2(x)
```

I ara ja es pot obtenir la representació gràfica:

```
(%i4) wxplot2d([F1(x)], [x,-6,6], [y, -6, 6],
               [gnuplot_preamble, "set size ratio 1; set grid; "], [legend, "F1(x)"], [style, [lines, 2, 2]], [ylabel, ""])$
plot2d: some values were clipped.
```

(%t4)



Si es considera la funció definida per:


$$G_1(x) = \begin{cases} e^{x/2}, & x \leq 0 \\ 1 + x \log(x/2), & 0 < x < 2 \\ e^{2-x}, & x \geq 2 \end{cases}$$

Aleshores:

```
(%i5) g1(x):=exp(x/2)$
      g2(x):=1+x*log(x/2)$
      g3(x):=exp(2-x)$
      G1(x):= if x<=0 then g1(x) else (if x<=2 then g2(x) else g3(x));
(%o8) G1(x):=if x<=0 then g1(x) else if x<=2 then g2(x) else g3(x)
```


```
(%i9) wxplot2d([G1(x)], [x,-6,6], [legend, "G1(x)"], [style, [lines, 2, 3]], [ylabel, ""])$
```

(%t9)



```
(%i10) wxplot2d([G1(x)], [x,-6,6], [y, 0, 1.4],  
[gnuplot_preamble, "set size ratio 1; set grid; "], [style, [lines, 2, 1]], [ylabel, ""]) $
```

(%t10)




Per a la representació gràfica de corbes planes donades per una equació $F(x,y)=0$, cal fer servir la instrucció `draw2d`, tal com es pot veure a continuació. Per a més detalls sobre les opcions, consulteu el manual del programa.

Representació gràfica de la circumferència de centre l'origen i radi unitat:

```
(%i49) load(draw)$
wxdraw2d(
  grid = true, line_type = solid, line_width = 2, color = blue,
  implicit(x^2+y^2-1=0, x, -3.0,3.0, y, -2,2)
);
```

(%t50)




(%o50) [gr2d(implicit)]

Representació gràfica de la circumferència de centre el punt (2,3) i radi 2:

```
(%i39) load(draw)$
wxdraw2d(
  grid = true, line_type = solid, line_width = 2, color = red,
  implicit((x-2)^2+(y-3)^2-4=0, x, -1,5, y, 0,6)
);
```

(%t40)




(%o40) [gr2d(implicit)]

Com es pot veure, la representació gràfica es fa per defecte en un format que pot semblar estrany, ja que l'escala dels eixos no és la mateixa i la corba pot aparèixer deformada. Per corregir aquest efecte es pot copiar la gràfica clicant amb el ratolí sobre

la gràfica i prement l'opció de copiar la imatge; un cop copiada i enganxada en un document, es pot aconseguir una escala que permeti una visualització més adient, com es pot veure amb la circumferència anterior.

Representació gràfica de la paràbola d'equació $x = y^2 + 2y + 1$:

```
(%i47) load(draw)$
      wxdraw2d(
        grid = true, line_type = solid, line_width = 2, color = green,
        implicit(y^2+2*y+1-x=0, x, -1,5, y, -4,2)
      );
```



```
(%o48) [gr2d(implicit)]
```

06-3.- Límits de funcions. Càlcul i propietats

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_06-3.wxm**.

Recordem la definició de límit d'una funció en un punt. Si $f: A \subset \mathbb{R} \rightarrow \mathbb{R}$ i $a \in \mathbb{R}$ és un punt d'acumulació del conjunt A , es diu que $L \in \mathbb{R}$ és el límit de la funció f en el punt $a \in \mathbb{R}$, si qualsevol que sigui $\varepsilon > 0$ existeix $\delta > 0$ tal que sempre que $x \in A$ i $0 < |x - a| < \delta$, aleshores es compleix $|f(x) - L| < \varepsilon$. Notació: $L = \lim_{x \rightarrow a} f(x)$ o bé $L = \lim_a f$. Si en la definició se substitueix la condició $0 < |x - a| < \delta$ per $a < x < a + \delta$, aleshores es té el límit lateral per la dreta de la funció en el punt i si se substitueix per $a - \delta < x < a$ aleshores es té el límit lateral per l'esquerra. El límit d'una funció en un punt existeix si i solament si, els límits laterals existeixen i són iguals.

El càlcul del límit d'una funció en un punt es duu a terme amb el menú "Anàlisis/Calcular límite". Quan s'aplica aquesta opció, s'obre el quadre de diàleg de la Figura 06-5. En aquest quadre de diàleg cal especificar:

- la referència de la funció o la seva expressió analítica;
- la variable;
- el punt en el qual es vol calcular el límit de la funció; hi ha una opció designada "Especial" en la que es pot indicar el nombre π , el nombre e , l'infinit o el menys infinit;
- si el límit es vol en el punt ("ambos lados") o lateral (esquerra o dreta).



Figura 06-5. Quadre de diàleg per calcular el límit d'una funció en un punt.

Per exemple, si es consideren les funcions

$$f_1(x) = \sin(x); \quad f_2(x) = 1 + \log(x+1)$$

plantejem el càlcul del límit de la funció f_2 en el punt $a = 1$ així com els límits laterals:

```
(%i1) f1(x):=sin(x)$    f2(x):=1+log(x+1)$
      L1:=limit(f2(x), x, 1);
      L2:=limit(f2(x), x, 1, minus);
      L3:=limit(f2(x), x, 1, plus);
(%o3) L1 = log(2)+1
(%o4) L2 = log(2)+1
(%o5) L3 = log(2)+1
```

Si es vol una sortida més completa, cal indicar a wxMaxima que escrigui el límit que es vol calcular, el signe “=” i a continuació la instrucció de càlcul del límit. Així, en el cas anterior el procediment seria:

```
(%i6) 'limit(f2(x), x, 1)=limit(f2(x), x, 1);
      'limit(f2(x), x, 1, minus)=limit(f2(x), x, 1, minus);
      'limit(f2(x), x, 1, plus)=limit(f2(x), x, 1, plus);
(%o6) lim log(x+1)+1=log(2)+1
      x→1
(%o7) lim log(x+1)+1=log(2)+1
      x→1-
(%o8) lim log(x+1)+1=log(2)+1
      x→1+
```

Considerem ara la funció definida a intervals per:

$$F_1(x) = \begin{cases} \sin(x), & \text{si } x \leq 0 \\ 1 + \log(x+1), & \text{si } x > 0 \end{cases}$$

i plantejem el càlcul del límit lateral per l'esquerra en el punt $a = 0$; la sintaxi que en primera instància utilitzaríem és:

```
(%i9) F1(x):= if x<=0 then f1(x) else f2(x);    'limit(F1(x), x, 0, minus)=limit(F1(x), x, 0, minus);
(%o9) F1(x):=if x≤0 then f1(x) else f2(x)
(%o10) lim (if x≤0 then f1(x) else f2(x))= lim if x≤0 then f1(x) else f2(x)
      x→0-                                x→0-
```


que, com es pot veure, no dóna el resultat desitjat. Això és perquè en aquest tipus de funcions cal escriure el símbol de cada una de les expressions analítiques que es fan servir en la definició de la funció definida a intervals.

Per tant, la sintaxi correcta és:

```
(%i11) 'limit(f1(x), x, 0, minus)=limit(f1(x), x, 0, minus);
      'limit(f2(x), x, 0, plus)=limit(f2(x), x, 0, plus);
(%o11) lim sin(x)=0
      x→0-
(%o12) lim log(x+1)+1=1
      x→0+
```

La representació gràfica que il·lustra aquesta situació obtinguda en wxMaxima és:

```
(%i13) wxplot2d([F1(x)], [x,-2,2],
      [gnuplot_preamble, "set grid;"], [ylabel, ""], [legend, "F1(x)"], [style, [lines, 2]] )$
```



Considerem ara la funció definida per:

$$f_3(x) = \frac{|x|}{x}, x \neq 0$$

El punt $a=0$ no és del camp d'existència d'aquesta funció, però és un punt d'acumulació d'aquest conjunt. Per tant, té sentit calcular el límit de la funció en aquest punt:

```
(%i14) f3(x):=(abs(x))/x;
(%o14) f3(x):=|x|/x
(%i15) 'limit(f3(x), x, 1)=limit(f3(x), x, 1);
(%o15) lim |x|/x = 1
      x→1
```

```
(%i16) 'limit(f3(x), x, 0)=limit(f3(x), x, 0);
```

```
(%o16)  $\lim_{x \rightarrow 0} \frac{|x|}{x} = \text{und}$ 
```

La resposta “und”, que és una abreviatura de “undefined” significa que el límit no existeix. Això no obstant, podem plantejar el càlcul dels límits laterals en aquest punt i observar que són diferents. La gràfica de la funció es representa amb la sortida que dona wxMaxima.

```
(%i17) 'limit(f3(x), x, 0, minus)=limit(f3(x), x, 0, minus);
```

```
(%o17)  $\lim_{x \rightarrow 0^-} \frac{|x|}{x} = -1$ 
```


```
(%i18) 'limit(f3(x), x, 0, plus)=limit(f3(x), x, 0, plus);
```

```
(%o18)  $\lim_{x \rightarrow 0^+} \frac{|x|}{x} = 1$ 
```

```
(%i19) wxplot2d([f3(x)], [x,-2,2], [y, -2, 2], [ylabel, ""], [legend, "f3(x)"], [style, [lines, 2]],  
[gnuplot_preamble, "set grid;"] )$
```

plot2d: expression evaluates to non-numeric value somewhere in plotting range.

```
(%t19)
```



Un comentari addicional sobre càlcul de límits de funcions: es poden calcular límits de funcions amb paràmetres en la seva expressió analítica i també en un punt genèric o arbitrari, tal com es mostra a continuació:

```
(%i20) f21(x):=(x+A)*(x^2+B);
```

```
(%o20) f21(x):=(x+A)(x^2+B)
```

```
(%i21) 'limit(f21(x), x, 1)=limit(f21(x), x, 1);
```

```
(%o21)  $\lim_{x \rightarrow 1} (A+x)(B+x^2) = (A+1)B + A + 1$ 
```

```
(%i22) 'limit(f21(x), x, a)=limit(f21(x), x, a);
(%o22)  $\lim_{x \rightarrow a} (A+x)(B+x^2) = (A+a)B + a^2 A + a^3$ 
```

Límits infinits. Com sabem, el límit d'una funció en un punt es generalitza a valors del límit de la recta real ampliada, és a dir, límits infinits. A l'exemple següent es mostra la resposta del programa en aquest cas:

```
(%i1) f1(x):=1/(x-1)^2;
(%o1)  $f1(x) := \frac{1}{(x-1)^2}$ 
(%i2) 'limit(f1(x), x, 1)=limit(f1(x), x, 1);
(%o2)  $\lim_{x \rightarrow 1} \frac{1}{(x-1)^2} = \infty$ 
```

Pot ser que en un cas la resposta sigui que el límit no existeix, però sí que existeixi un dels límits laterals i sigui infinit, com es pot veure a l'exemple següent:

```
(%i3) 'limit(log(x), x, 0)=limit(log(x), x, 0);
      'limit(log(x), x, 0, plus)=limit(log(x), x, 0, plus);
(%o3)  $\lim_{x \rightarrow 0} \log(x) = \text{infinity}$ 
(%o4)  $\lim_{x \rightarrow 0+} \log(x) = -\infty$ 
```

Recordem que quan una funció té límit infinit o algun dels límits laterals infinits en un punt a , la recta d'equació $x = a$ es diu que és una asymptota vertical de la gràfica de la funció.

En ocasions la resposta del programa pot resultar sorprenent, però en cap cas cal que sigui enganyosa en la seva interpretació. Tal és el cas de la funció $f(x) = 1/x$, $x \neq 0$ a l'origen:

```
(%i5) f2(x):=1/x;
(%o5)  $f2(x) := \frac{1}{x}$ 
(%i6) 'limit(f2(x), x, 0)=limit(f2(x), x, 0);
(%o6)  $\lim_{x \rightarrow 0} \frac{1}{x} = \text{infinity}$ 
```

```
(%i7) 'limit(f2(x), x, 0, plus)=limit(f2(x), x, 0, plus);
(%o7)  $\lim_{x \rightarrow 0^+} \frac{1}{x} = \infty$ 
(%i8) 'limit(f2(x), x, 0, minus)=limit(f2(x), x, 0, minus);
(%o8)  $\lim_{x \rightarrow 0^-} \frac{1}{x} = -\infty$ 
```

Límits a l'infinit. Els límits a $+\infty$ i a $-\infty$ es calculen amb una sintaxi similar a la de la instrucció de càlcul del límit d'una funció, especificant “inf” i “minf”, respectivament; per exemple:

```
(%i9) 'limit(exp(x), x, minf)=limit(exp(x), x, minf);
      'limit(exp(x), x, inf)=limit(exp(x), x, inf);
(%o9)  $\lim_{x \rightarrow -\infty} e^x = 0$ 
(%o10)  $\lim_{x \rightarrow \infty} e^x = \infty$ 
(%i11) 'limit(2-exp(-x), x, minf)=limit(2-exp(-x), x, minf);
      'limit(2-exp(-x), x, inf)=limit(2-exp(-x), x, inf);
(%o11)  $\lim_{x \rightarrow -\infty} 2 - e^{-x} = -\infty$ 
(%o12)  $\lim_{x \rightarrow \infty} 2 - e^{-x} = 2$ 
```

Com es pot veure, en alguns casos un d'aquests límits pot ser també infinit. Quan el límit a l'infinit dóna un valor finit L , la recta d'equació $y = L$ es diu que és una asymptota horitzontal de la gràfica de la funció, que ho serà a la banda de la recta real corresponent.

Hi ha funcions que tenen límits finits a les dues bandes de l'infinit de la recta real, com per exemple la funció definida per:

$$g_1(x) = 1 + \frac{1}{1+x^2}$$

```
(%i13) g1(x):=1+1/(1+x^2);
(%o13)  $g_1(x) := 1 + \frac{1}{1+x^2}$ 
(%i14) 'limit(g1(x), x, minf)=limit(g1(x), x, minf);
(%o14)  $\lim_{x \rightarrow -\infty} \frac{1}{x^2+1} + 1 = 1$ 
```

```
(%i15) 'limit(g1(x), x, inf)=limit(g1(x), x, inf);
```

```
(%o15)  $\lim_{x \rightarrow \infty} \frac{1}{x^2+1} + 1 = 1$ 
```

Finalment, el límit a l'infinit es pot aplicar al càlcul de les asímptotes obliques d'una funció, que com se sap són rectes que compleixen la condició

$$\lim_{\infty} (f(x) - (mx + b)) = 0$$

El càlcul dels coeficients de l'asímtota es duu a terme mitjançant les expressions:

$$m = \lim_{\infty} \frac{f(x)}{x}; \quad b = \lim_{\infty} (f(x) - mx)$$

Calculem, per exemple, l'asímtota obliqua de la funció $f(x) = x + \frac{1}{x}$, $x \neq 0$:

```
(%i16) g2(x):=x+1/x;
```

```
      m1=m1:limit(g2(x)/x, x, minf);
```

```
      m2=m2:limit(g2(x)/x, x, inf);
```

```
(%o16)  $g2(x) := x + \frac{1}{x}$ 
```

```
(%o17) m1 = 1
```

```
(%o18) m2 = 1
```

```
(%i19) b1=b1:limit((g2(x)-m1*x), x, minf);
```

```
      b2=b2:limit((g2(x)-m2*x), x, inf);
```

```
(%o19) b1 = 0
```

```
(%o20) b2 = 0
```

```
(%i21) y(x):=m1*x+b1; 'y=y(x);
```

```
(%o21)  $y(x) := m1 x + b1$ 
```

```
(%o22)  $y = x$ 
```


La representació gràfica que dona wxMaxima de la gràfica d'aquesta funció és la que es pot veure a continuació.

```
(%i23) wxplot2d([g2(x),y(x)], [x,-6,6], [y,-6,6], [legend, "g2(x)", "y(x)"], [style, [lines, 2]] ,
[gnuplot_preamble, "set grid;"]
)$
```

plot2d: expression evaluates to non-numeric value somewhere in plotting range.

plot2d: some values were clipped.

(%t23)



06-4.- Funcions contínues. Estudi de la continuïtat

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_06-4.wxm**.

Recordem la definició de funció contínua en un punt. Si $f : A \subset \mathbb{R} \rightarrow \mathbb{R}$ i $a \in A$ és un punt no aïllat, es diu que la funció és contínua en aquest punt si es compleix la condició $\lim_{x \rightarrow a} f(x) = f(a)$. Aquesta definició implica que

- el límit de la funció en el punt existeix i és finit;
- el límit de la funció en el punt és igual a la imatge del punt per la funció.

Si una funció no és contínua en un punt es diu que és discontinua. Les discontinuïtats es classifiquen en:

- discontinuïtat evitable: si el límit existeix però és diferent de la imatge;
- discontinuïtat de salt: si els límits laterals existeixen, són finits i diferents;
- discontinuïtat essencial: quan un dels límits laterals o tots dos no existeixen o no són finits.

L'estudi de la continuïtat s'ha de fer aplicant els criteris de continuïtat de funcions i, en tot cas, ajudant-se o complementant l'estudi amb uns càlculs de límits o representacions gràfiques. No hi ha cap instrucció de wxMaxima que doni una resposta a una possible pregunta sobre la continuïtat d'una funció en un punt.

- Exemple de discontinuïtat evitable: la funció definida per

$$f(x) = \begin{cases} \exp(x), & \text{si } x \neq 0 \\ 2, & \text{si } x = 0 \end{cases}$$

La gràfica es pot veure a la figura 06-6. És immediat verificar que es compleix:

$$\lim_{x \rightarrow 0^-} f(x) = \lim_{x \rightarrow 0^+} f(x) = \lim_{x \rightarrow 0} f(x) = 1 \neq f(0) = 2$$




Figura 06-6. Il·lustració de la discontinuïtat evitable

- Exemple de discontinuïtat de salt: la funció definida per


$$f(x) = \begin{cases} x^2, & \text{si } x < 0 \\ x^2 + 2, & \text{si } x \geq 0 \end{cases}$$

En wxMaxima la funció es defineix mitjançant:

```
(%i1) f1(x):=x^2; f2(x):=x^2+2; f(x):= if x<0 then f1(x) else f2(x);
(%o1) f1(x):=x^2
(%o2) f2(x):=x^2+2
(%o3) f(x):=if x<0 then f1(x) else f2(x)
```

La representació gràfica de la funció es pot obtenir en wxMaxima:

```
(%i4) wxplot2d([f(x)], [x,-3,3], [y,-2,12], [ylabel, ""], [style, [lines, 2]], [gnuplot_preamble, "set grid;"]);
```



És immediat verificar que es compleix:

$$\lim_{x \rightarrow 0^-} f(x) = 0; \quad \lim_{x \rightarrow 0^+} f(x) = 2$$

- Exemple de discontinuïtat essencial: la funció definida per


$$f(x) = \begin{cases} x^2, & \text{si } x \leq 0 \\ \sin(1/x), & \text{si } x > 0 \end{cases}$$

En wxMaxima la funció es defineix mitjançant:

```
(%i5) f3(x):=sin(1/x);  g(x):= if x<=0 then f1(x) else f3(x);
(%o5) f3(x):=sin( $\frac{1}{x}$ )
(%o6) g(x):=if x≤0 then f1(x) else f3(x)
```

La representació gràfica de la funció es pot obtenir en wxMaxima:

```
(%i7) wxplot2d([g(x)], [x,-1,1], [y,-1.5,1.5], [ylabel, ""], [style, [lines, 2]],
[gnuplot_preamble, "set grid;"] )$
```



És immediat verificar que es compleix:

$$\lim_{x \rightarrow 0^-} f(x) = 0; \quad \lim_{x \rightarrow 0^+} f(x) = \text{no existeix}$$

06-5.- Propietats de les funcions contínues. Aplicacions

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_06-5.wxm**.

Les propietats més importants de les funcions contínues en un interval tancat són:

- el teorema de Bolzano;
- el teorema dels valors intermedis;
- el teorema del màxim i del mínim;
- el teorema de Weierstrass.

El teorema de Bolzano permet afirmar l'existència d'almenys una arrel o zero d'una funció contínua en un interval tancat, si la funció té signe diferent en els extrems. Aquest teorema permet calcular arrels d'equacions que amb la instrucció "solve" no es poden determinar. Per exemple, es planteja calcular les arrels de l'equació


$$\exp(-x) - x = 0$$

En primer lloc es defineix la funció $f(x) = e^{-x} - x$, $x \in \mathbb{R}$. Observem que aquesta funció és contínua en el seu camp d'existència. Ara es planteja resoldre l'equació $f(x)=0$:

```
(%i1) f(x):=exp(-x)-x;      Eq1:f(x)=0;
      solve([Eq01], [x]);
(%o1) f(x):=exp(-x)-x
      (%o2) %e-x-x=0
      (%o3) []
```


Evidentment s'observa un problema: wxMaxima no dona cap resposta i, per tant, cal buscar una alternativa. Aquesta passa per observar que la funció canvia de signe a l'interval tancat $[-3,3]$, tal com es pot veure a la gràfica que s'obté en wxMaxima:

```
(%i4) wxplot2d([f(x)], [x,-3,3], [ylabel, ""], [style, [lines, 2]], [gnuplot_preamble, "set grid;"] )$
```



Es pot “afinar més” i afirmar que la funció canvia de signe en l’interval $[0,1]$, centrant la representació gràfica en aquest interval:

```
(%i5) wxplot2d([f(x)], [x,0,1], [ylabel, ""], [style, [lines, 2]], [gnuplot_preamble, "set grid;"])$
```



Per tant, en virtut del Teorema de Bolzano, es pot afirmar que hi ha una solució en aquest interval. En el menú “Ecuaciones” hi ha una opció “find root” que aplica un algorisme numèric d’aproximació de la solució d’una equació o d’una funció en un interval tancat. En aquest cas s’obté:

```
(%i6) find_root(Eq1, x, 0, 2);
(%o6) 0.56714329040978
```

Cal tenir present que aquesta opció funciona correctament quan a l’interval tancat que s’especifica per a trobar el zero, la funció té només una arrel; en cas contrari dóna la major de les arrels. Per exemple, si es considera la funció

$$g(x) = x^2 \sin\left(\frac{1}{x}\right), x > 0$$

i s’observa la seva gràfica a l’interval $[0.05; 0.2]$, la resposta de wxMaxima al càlcul de l’arrel és un únic resultat, que com es pot veure al gràfic correspon al major dels zeros de la funció en aquest interval:

```
(%i8) g(x):=x^2*sin(1/x);      Eq2:g(x)=0;
      find_root(Eq2, x, 0.05, 0.2);
```

```
(%o8) g(x):=x^2 sin(1/x)
```

```
(%o9) sin(1/x) x^2 = 0
```

```
(%o10) 0.1591549430919
```

Així mateix, cal assegurar-se que la funció canvia de signe en els extrems de l'interval en el qual es busca el zero; en cas contrari, wxMaxima ho adverteix i no fa cap càlcul. Per exemple, si s'observa la gràfica anterior, es veu que la funció té dos zeros a l'interval $[0.1; 0.2]$; tanmateix, la resposta de wxMaxima és:

```
(%i11) find_root(Eq2, x, 0.1, 0.2);
find_root: function has same sign at endpoints: f(0.1)=-0.0054402111088937, f(0.2)=-0.038356970986526
-- an error. To debug this try: debugmode(true);
```

Si s'estableix un interval en el qual hi ha una única arrel, s'obté la resposta correcta:

```
(%i12) find_root(Eq2, x, 0.14, 0.2);
(%o12) 0.1591549430919
```

Representació gràfica de la funció en $[0.05, 0.20]$:

```
(%i13) wxplot2d([g(x)], [x,0.05,0.2], [ylabel, ""], [style, [lines, 2]], [gnuplot_preamble, "set grid;"])$
```

(%t13)

